
G13 – Time Series Analysis

G13ACF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G13ACF calculates partial autocorrelation coefficients given a set of autocorrelation coefficients. It also
calculates the predictor error variance ratios for increasing order of finite lag autoregressive predictor,
and the autoregressive parameters associated with the predictor of maximum order.

2 Specification

SUBROUTINE G13ACF(R, NK, NL, P, V, AR, NVL, IFAIL)
INTEGER NK, NL, NVL, IFAIL
real R(NK), P(NL), V(NL), AR(NL)

3 Description

The data consist of values of autocorrelation coefficients r1, r2, . . . , rK , relating to lags 1, 2, . . . , K. These
will generally (but not necessarily) be sample values such as may be obtained from a time series xt using
G13ABF.

The partial autocorrelation coefficient at lag l may be identified with the parameter pl,l in the
autoregression

xt = cl + pl,1xt−1 + pl,2xt−2 + . . .+ pl,lxt−l + el,t

where el,t is the predictor error.

The first subscript l of pl,l and el,t emphasises the fact that the parameters will in general alter as further
terms are introduced into the equation (i.e., as l is increased).

The parameters are determined from the autocorrelation coefficients by the Yule–Walker equations

ri = pl,1ri−1 + pl,2ri−2 + . . .+ pl,lri−l, i = 1, 2, . . . , l

taking rj = r|j| when j < 0, and r0 = 1.

The predictor error variance ratio vl = var(el,t)/ var(xt) is defined by vl = 1− pl,1r1 − pl,2r2 − . . .− pl,lrl.

The above sets of equations are solved by a recursive method (the Durbin–Levinson algorithm). The
recursive cycle applied for l = 1, 2, . . . , (L−1), where L is the number of partial autocorrelation coefficients
required, is initialised by setting p1,1 = r1 and v1 = 1− r2

1 .

Then
pl+1,l+1 = (rl+1 − pl,1rl − pl,2rl−1 − . . . − pl,lr1)/vl

pl+1,j = pl,j − pl+1,l+1pl,l+1−j , j = 1, 2, . . . , l
vl+1 = vl(1− pl+1,l+1)(1 + pl+1,l+1).

If the condition |pl,l| ≥ 1 occurs, say when l = l0, it indicates that the supplied autocorrelation coefficients
do not form a positive-definite sequence (see Hannan [3]), and the recursion is not continued. The
autoregressive parameters are overwritten at each recursive step, so that upon completion the only
available values are pLj, for j = 1, 2, . . . , L or pl0−1,j if the recursion has been prematurely halted.

4 References

5 Parameters

1: R(NK) — real array Input

On entry: the autocorrelation coefficient relating to lag k, for k = 1, 2, . . . , K.

2: NK — INTEGER Input

On entry: the number of lags, K. The lags range from 1 to K and do not include zero.

Constraint: NK > 0.

3: NL — INTEGER Input

On entry: the number of partial autocorrelation coefficients required, L.

Constraint: 0 < NL ≤ NK.

4: P(NL) — real array Output

On exit: P(l) contains the partial autocorrelation coefficient at lag l, pl,l, for l = 1, 2, . . . ,NVL.

5: V(NL) — real array Output

On exit: V(l) contains the predictor error variance ratio vl, for l = 1, 2, . . . ,NVL.

6: AR(NL) — real array Output

On exit: the autoregressive parameters of maximum order, i.e., pLj if IFAIL = 0, or pl0−1,j if
IFAIL = 3, for j = 1, 2, . . . ,NVL.

7: NVL — INTEGER Output

On exit: the number of valid values in each of P, V and AR. Thus in the case of premature
termination at iteration l0 (see Section 3), NVL is returned as l0 − 1.

8: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, NK ≤ 0,

or NL ≤ 0,

or NK < NL.

IFAIL = 2

On entry, the autocorrelation coefficient of lag 1 has an absolute value greater than or equal to
1.0; no recursions could be performed.

IFAIL = 3

Recursion has been prematurely terminated; the supplied autocorrelation coefficients do not form
a positive-definite sequence (see Section 3). Parameter NVL returns the number of valid values
computed.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by the routine is proportional to (NVL)2.

9 Example

In the example below the input series is the set of 10 sample autocorrelation coefficients derived from
the original series of sunspot numbers by G13ABF example program. The results show 5 values of
each of the three output arrays – partial autocorrelation coefficients, predictor error variance ratios and
autoregressive parameters. All of these were valid.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G13ACF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NLMAX, NKMAX
PARAMETER (NLMAX=5,NKMAX=10)
INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)

* .. Local Scalars ..
INTEGER I, IFAIL, NK, NL, NVL

* .. Local Arrays ..
real AR(NLMAX), P(NLMAX), R(NKMAX), V(NLMAX)

* .. External Subroutines ..
EXTERNAL G13ACF

* .. Executable Statements ..
WRITE (NOUT,*) ’G13ACF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) NK, NL
WRITE (NOUT,*)
IF (NL.GT.0 .AND. NL.LE.NLMAX .AND. NK.GT.0 .AND. NK.LE.NKMAX)

+ THEN
READ (NIN,*) (R(I),I=1,NK)
IFAIL = 1

*
CALL G13ACF(R,NK,NL,P,V,AR,NVL,IFAIL)

*
IF (IFAIL.NE.0) THEN

WRITE (NOUT,99999) ’G13ACF fails. IFAIL = ’, IFAIL
WRITE (NOUT,*)

END IF
IF (IFAIL.EQ.3) THEN

WRITE (NOUT,99998) ’ Only’, NVL,
+ ’valid sets were generated’

WRITE (NOUT,*)
END IF
IF (IFAIL.EQ.0 .OR. IFAIL.EQ.3) THEN

WRITE (NOUT,*)
+ ’Lag Partial Predictor error Autoregressive’

WRITE (NOUT,*) ’ autocorrn variance ratio parameter’
WRITE (NOUT,*)
WRITE (NOUT,99997) (I,P(I),V(I),AR(I),I=1,NVL)

END IF

END IF
STOP

*
99999 FORMAT (1X,A,I1)
99998 FORMAT (1X,A,I2,A)
99997 FORMAT (1X,I2,F9.3,F16.3,F14.3)

END

9.2 Program Data

G13ACF Example Program Data
10 5

0.8004 0.4355 0.0328 -0.2835 -0.4505
-0.4242 -0.2419 -0.0550 0.3783 0.5857

9.3 Program Results

G13ACF Example Program Results

Lag Partial Predictor error Autoregressive
autocorrn variance ratio parameter

1 0.800 0.359 1.108
2 -0.571 0.242 -0.290
3 -0.239 0.228 -0.193
4 -0.049 0.228 -0.014
5 -0.032 0.228 -0.032

